

# The Penultimate Alexander Polynomial

September 2008, Joint with Jana Archibald

```

In[ ]:= Print[prog = Import[
  "C:\\drorbn\\AcademicPensieve\\Talks\\Sandbjerg-0810/pA.txt",
  "Text"
]];
ToExpression[prog]

(* WP: Wedge Product *)
WSort[expr_] := Expand[expr /. w_W -> Signature[w]*Sort[w]];
WP[0, _] = WP[_ , 0] = 0;
WP[a_, b_] := WSort[Distribute[a ** b] /.
  (c1_. * w1_W) ** (c2_. * w2_W) -> c1 c2 Join[w1, w2]];

(* IM: Interior Multiplication *)
IM[{}, expr_] := expr;
IM[i_, w_W] := If[FreeQ[w, i], 0,
  -(-1)^Position[w, i][[1,1]]*DeleteCases[w, i]];
IM[{is___, i_}, w_W] := IM[{is}, IM[i, w]];
IM[is_list, expr_] := expr /. w_W -> IM[is, w]

(* pA on Crossings *)
pA[Xp[i_, j_, k_, l_]] := AHD[(t[i]==t[k])(t[j]==t[l]), {i,l}, W[j,k],
  W[l,i] + (t[i]-1)W[l,j] - t[l]W[l,k] + W[i,j] + t[l]W[j,k]];
pA[Xm[i_, j_, k_, l_]] := AHD[(t[i]==t[k])(t[j]==t[l]), {i,j}, W[k,l],
  t[j]W[i,j] - t[j]W[i,l] + W[j,k] + (t[i]-1)W[j,l] + W[k,l]]

(* Variable Equivalences *)
ReductionRules[Times[]] = {};
ReductionRules[Equal[a_, b_]] := (# -> a)& /@ {b};
ReductionRules[eqs_Times] := Join @@ (ReductionRules /@ List@@eqs)


(* AHD: Alexander Half Densities *)
AHD[eqs_, is_, -os_, p_] := AHD[eqs, is, os, Expand[-p]];
AHD /: Reduce[AHD[eqs_, is_, os_, p_]] :=
  AHD[eqs, Sort[is], WSort[os], WSort[p /. ReductionRules[eqs]]];
AHD /: AHD[eqs1_, is1_, os1_, p1_] AHD[eqs2_, is2_, os2_, p2_] := Module[
  {glued = Intersection[Union[is1, is2], List@@Union[os1, os2]]},
  Reduce[AHD[
    eqs1*eqs2 /. eq1_Equal*eq2_Equal /;
    Intersection[List@@eq1, List@@eq2] != {} -> Union[eq1, eq2],
    Complement[Union[is1, is2], glued],
    IM[glued, WP[os1, os2]],
    IM[glued, WP[p1, p2]]
  ] ] ]

(* pA on Circuit Diagrams *)
pA[cd_CircuitDiagram, eqs___] := pA[cd, {}, AHD[Times[eqs], {}, W[], W[]]];
pA[cd_CircuitDiagram, done_, ahd_AHD] := Module[
  {pos = First[Ordering[Length[Complement[List@@#, done]] & /@ cd]]},
  pA[Delete[cd, pos], Union[done, List@@cd[[pos]]], ahd*pA[cd[[pos]]]]
];
pA[CircuitDiagram[], _, ahd_AHD] := ahd

```

## Some Relations

### Reidemeister 3




```
In[ ]:= res1 = pA /@ {
  CircuitDiagram[Xp[7, 9, 6, 1], Xp[3, 8, 7, 2], Xp[8, 4, 5, 9]],
  CircuitDiagram[Xp[3, 4, 7, 9], Xp[7, 5, 6, 8], Xp[2, 9, 8, 1]]
}
```

```
Out[ ]:= {AHD[(t[1] == t[4] == t[9]) (t[2] == t[5] == t[8]) (t[3] == t[6] == t[7]),
  {1, 2, 3}, W[4, 5, 6], -W[1, 2, 3] + W[1, 2, 4] - t[3] W[1, 2, 4] + t[1] W[1, 2, 5] -
  t[1] t[3] W[1, 2, 5] + t[1] t[2] W[1, 2, 6] - W[1, 3, 4] + t[2] W[1, 3, 4] -
  t[1] W[1, 3, 5] + t[1] t[2] W[1, 4, 5] - t[1] t[2] t[3] W[1, 4, 5] - t[1] t[2] W[1, 4, 6] +
  t[1] t[2]^2 W[1, 4, 6] - t[1]^2 t[2] W[1, 5, 6] + W[2, 3, 4] + t[1] W[2, 4, 5] -
  t[1] t[3] W[2, 4, 5] + t[1] t[2] W[2, 4, 6] - t[1] W[3, 4, 5] + t[1]^2 t[2] W[4, 5, 6]],
  AHD[(t[1] == t[4] == t[9]) (t[2] == t[5] == t[8]) (t[3] == t[6] == t[7]), {1, 2, 3},
  W[4, 5, 6], -W[1, 2, 3] + W[1, 2, 4] - t[3] W[1, 2, 4] + t[1] W[1, 2, 5] -
  t[1] t[3] W[1, 2, 5] + t[1] t[2] W[1, 2, 6] - W[1, 3, 4] + t[2] W[1, 3, 4] -
  t[1] W[1, 3, 5] + t[1] t[2] W[1, 4, 5] - t[1] t[2] t[3] W[1, 4, 5] - t[1] t[2] W[1, 4, 6] +
  t[1] t[2]^2 W[1, 4, 6] - t[1]^2 t[2] W[1, 5, 6] + W[2, 3, 4] + t[1] W[2, 4, 5] -
  t[1] t[3] W[2, 4, 5] + t[1] t[2] W[2, 4, 6] - t[1] W[3, 4, 5] + t[1]^2 t[2] W[4, 5, 6]]}
```

```
In[ ]:= Equal @@ res1
```

```
Out[ ]:= True
```


### Overcrossings Commute



```
In[ ]:= Equal[
  pA[CircuitDiagram[Xp[1, 7, 4, 6], Xp[2, 3, 5, 7]]],
  pA[CircuitDiagram[Xp[2, 7, 5, 6], Xp[1, 3, 4, 7]]] ]
```

```
Out[ ]:= True
```

## Commutators Commute



```
In[ ]:= Equal[
  pA[CircuitDiagram[Xp[1, 2, 11, 8], Xm[11, 3, 12, 7], Xp[12, 4, 13, 10], Xm[13, 5, 6, 9]],
  t[2] == t[3], t[4] == t[5]],
  pA[CircuitDiagram[Xp[1, 4, 11, 10], Xm[11, 5, 12, 9], Xp[12, 2, 13, 8], Xm[13, 3, 6, 7]],
  t[2] == t[3], t[4] == t[5]] ]
```

Out[ ]:= True

```
In[ ]:= res2 = pA[#, t[2] == t[3], t[4] == t[5]] & /@ {
  CircuitDiagram[Xp[1, 2, 11, 8], Xm[11, 3, 12, 7], Xp[12, 4, 13, 10], Xm[13, 5, 6, 9]],
  CircuitDiagram[Xp[1, 4, 11, 10], Xm[11, 5, 12, 9], Xp[12, 2, 13, 8], Xm[13, 3, 6, 7]]
}
```

```
Out[ ]:= {AHD[(t[2] == t[3] == t[7] == t[8]) (t[4] == t[5] == t[9] == t[10])
  (t[1] == t[6] == t[11] == t[12] == t[13]), {1, 3, 5, 8, 10}, W[2, 4, 6, 7, 9],
  t[2] t[4] W[1, 2, 3, 4, 5] - t[2] t[4] W[1, 2, 3, 4, 9] + t[2] t[4] W[1, 2, 3, 5, 10] -
  t[2] t[4] W[1, 2, 3, 9, 10] - t[2] t[4] W[1, 2, 4, 5, 7] - t[2] t[4] W[1, 2, 4, 7, 9] +
  t[2] t[4] W[1, 2, 5, 7, 10] + t[2] t[4] W[1, 2, 7, 9, 10] + t[2] t[4] W[1, 3, 4, 5, 8] +
  t[2] t[4] W[1, 3, 4, 8, 9] - t[2] t[4] W[1, 3, 5, 8, 10] - t[2] t[4] W[1, 3, 8, 9, 10] -
  t[2] t[4] W[1, 4, 5, 7, 8] + t[2] t[4] W[1, 4, 7, 8, 9] - t[2] t[4] W[1, 5, 7, 8, 10] +
  t[2] t[4] W[1, 7, 8, 9, 10] - t[2] t[4] W[2, 3, 4, 5, 6] + t[2] t[4] W[2, 3, 4, 5, 7] -
  t[1] t[2] t[4] W[2, 3, 4, 5, 7] - t[2] t[4] W[2, 3, 4, 5, 8] + t[1] t[2] t[4] W[2, 3, 4, 5, 8] +
  t[2] t[4] W[2, 3, 4, 5, 9] - t[1] t[2] t[4] W[2, 3, 4, 5, 9] - t[2] t[4] W[2, 3, 4, 5, 10] +
  t[1] t[2] t[4] W[2, 3, 4, 5, 10] - t[2] t[4] W[2, 3, 4, 6, 9] + t[2] t[4] W[2, 3, 4, 7, 9] -
  t[1] t[2] t[4] W[2, 3, 4, 7, 9] - t[2] t[4] W[2, 3, 4, 8, 9] + t[1] t[2] t[4] W[2, 3, 4, 8, 9] +
  t[2] t[4] W[2, 3, 4, 9, 10] - t[1] t[2] t[4] W[2, 3, 4, 9, 10] +
  t[2] t[4] W[2, 3, 5, 6, 10] - t[2] t[4] W[2, 3, 5, 7, 10] + t[1] t[2] t[4] W[2, 3, 5, 7, 10] +
  t[2] t[4] W[2, 3, 5, 8, 10] - t[1] t[2] t[4] W[2, 3, 5, 8, 10] -
  t[2] t[4] W[2, 3, 5, 9, 10] + t[1] t[2] t[4] W[2, 3, 5, 9, 10] + t[2] t[4] W[2, 3, 6, 9, 10] -
  t[2] t[4] W[2, 3, 7, 9, 10] + t[1] t[2] t[4] W[2, 3, 7, 9, 10] + t[2] t[4] W[2, 3, 8, 9, 10] -
  t[1] t[2] t[4] W[2, 3, 8, 9, 10] - t[2] t[4] W[2, 4, 5, 6, 7] + t[2] t[4] W[2, 4, 5, 7, 8] -
  t[1] t[2] t[4] W[2, 4, 5, 7, 8] - t[2] t[4] W[2, 4, 5, 7, 9] + t[1] t[2] t[4] W[2, 4, 5, 7, 9] +
  t[2] t[4] W[2, 4, 5, 7, 10] - t[1] t[2] t[4] W[2, 4, 5, 7, 10] + t[2] t[4] W[2, 4, 6, 7, 9] -
  t[2] t[4] W[2, 4, 7, 8, 9] + t[1] t[2] t[4] W[2, 4, 7, 8, 9] + t[2] t[4] W[2, 4, 7, 9, 10] -
  t[1] t[2] t[4] W[2, 4, 7, 9, 10] - t[2] t[4] W[2, 5, 6, 7, 10] +
  t[2] t[4] W[2, 5, 7, 8, 10] - t[1] t[2] t[4] W[2, 5, 7, 8, 10] - t[2] t[4] W[2, 5, 7, 9, 10] +
  t[1] t[2] t[4] W[2, 5, 7, 9, 10] + t[2] t[4] W[2, 6, 7, 9, 10] - t[2] t[4] W[2, 7, 8, 9, 10] +
  t[1] t[2] t[4] W[2, 7, 8, 9, 10] + t[2] t[4] W[3, 4, 5, 6, 8] - t[2] t[4] W[3, 4, 5, 7, 8] +
  t[1] t[2] t[4] W[3, 4, 5, 7, 8] + t[2] t[4] W[3, 4, 5, 8, 9] - t[1] t[2] t[4] W[3, 4, 5, 8, 9] -
  t[2] t[4] W[3, 4, 5, 8, 10] + t[1] t[2] t[4] W[3, 4, 5, 8, 10] - t[2] t[4] W[3, 4, 6, 8, 9] +
  t[2] t[4] W[3, 4, 7, 8, 9] - t[1] t[2] t[4] W[3, 4, 7, 8, 9] - t[2] t[4] W[3, 4, 8, 9, 10] +
  t[1] t[2] t[4] W[3, 4, 8, 9, 10] + t[2] t[4] W[3, 5, 6, 8, 10] -
```

```

t[2] t[4] W[3, 5, 7, 8, 10] + t[1] t[2] t[4] W[3, 5, 7, 8, 10] +
t[2] t[4] W[3, 5, 8, 9, 10] - t[1] t[2] t[4] W[3, 5, 8, 9, 10] - t[2] t[4] W[3, 6, 8, 9, 10] +
t[2] t[4] W[3, 7, 8, 9, 10] - t[1] t[2] t[4] W[3, 7, 8, 9, 10] + t[2] t[4] W[4, 5, 6, 7, 8] -
t[2] t[4] W[4, 5, 7, 8, 9] + t[1] t[2] t[4] W[4, 5, 7, 8, 9] + t[2] t[4] W[4, 5, 7, 8, 10] -
t[1] t[2] t[4] W[4, 5, 7, 8, 10] + t[2] t[4] W[4, 6, 7, 8, 9] - t[2] t[4] W[4, 7, 8, 9, 10] +
t[1] t[2] t[4] W[4, 7, 8, 9, 10] - t[2] t[4] W[5, 6, 7, 8, 10] + t[2] t[4] W[5, 7, 8, 9, 10] -
t[1] t[2] t[4] W[5, 7, 8, 9, 10] - t[2] t[4] W[6, 7, 8, 9, 10] ],
AHD[(t[2] == t[3] == t[7] == t[8]) (t[4] == t[5] == t[9] == t[10])
(t[1] == t[6] == t[11] == t[12] == t[13]), {1, 3, 5, 8, 10}, W[2, 4, 6, 7, 9],
t[2] t[4] W[1, 2, 3, 4, 5] - t[2] t[4] W[1, 2, 3, 4, 9] + t[2] t[4] W[1, 2, 3, 5, 10] -
t[2] t[4] W[1, 2, 3, 9, 10] - t[2] t[4] W[1, 2, 4, 5, 7] - t[2] t[4] W[1, 2, 4, 7, 9] +
t[2] t[4] W[1, 2, 5, 7, 10] + t[2] t[4] W[1, 2, 7, 9, 10] + t[2] t[4] W[1, 3, 4, 5, 8] +
t[2] t[4] W[1, 3, 4, 8, 9] - t[2] t[4] W[1, 3, 5, 8, 10] - t[2] t[4] W[1, 3, 8, 9, 10] -
t[2] t[4] W[1, 4, 5, 7, 8] + t[2] t[4] W[1, 4, 7, 8, 9] - t[2] t[4] W[1, 5, 7, 8, 10] +
t[2] t[4] W[1, 7, 8, 9, 10] - t[2] t[4] W[2, 3, 4, 5, 6] + t[2] t[4] W[2, 3, 4, 5, 7] -
t[1] t[2] t[4] W[2, 3, 4, 5, 7] - t[2] t[4] W[2, 3, 4, 5, 8] + t[1] t[2] t[4] W[2, 3, 4, 5, 8] +
t[2] t[4] W[2, 3, 4, 5, 9] - t[1] t[2] t[4] W[2, 3, 4, 5, 9] - t[2] t[4] W[2, 3, 4, 5, 10] +
t[1] t[2] t[4] W[2, 3, 4, 5, 10] - t[2] t[4] W[2, 3, 4, 6, 9] + t[2] t[4] W[2, 3, 4, 7, 9] -
t[1] t[2] t[4] W[2, 3, 4, 7, 9] - t[2] t[4] W[2, 3, 4, 8, 9] + t[1] t[2] t[4] W[2, 3, 4, 8, 9] +
t[2] t[4] W[2, 3, 4, 9, 10] - t[1] t[2] t[4] W[2, 3, 4, 9, 10] +
t[2] t[4] W[2, 3, 5, 6, 10] - t[2] t[4] W[2, 3, 5, 7, 10] + t[1] t[2] t[4] W[2, 3, 5, 7, 10] +
t[2] t[4] W[2, 3, 5, 8, 10] - t[1] t[2] t[4] W[2, 3, 5, 8, 10] -
t[2] t[4] W[2, 3, 5, 9, 10] + t[1] t[2] t[4] W[2, 3, 5, 9, 10] + t[2] t[4] W[2, 3, 6, 9, 10] -
t[2] t[4] W[2, 3, 7, 9, 10] + t[1] t[2] t[4] W[2, 3, 7, 9, 10] + t[2] t[4] W[2, 3, 8, 9, 10] -
t[1] t[2] t[4] W[2, 3, 8, 9, 10] - t[2] t[4] W[2, 4, 5, 6, 7] + t[2] t[4] W[2, 4, 5, 7, 8] -
t[1] t[2] t[4] W[2, 4, 5, 7, 8] - t[2] t[4] W[2, 4, 5, 7, 9] + t[1] t[2] t[4] W[2, 4, 5, 7, 9] +
t[2] t[4] W[2, 4, 5, 7, 10] - t[1] t[2] t[4] W[2, 4, 5, 7, 10] + t[2] t[4] W[2, 4, 6, 7, 9] -
t[2] t[4] W[2, 4, 7, 8, 9] + t[1] t[2] t[4] W[2, 4, 7, 8, 9] + t[2] t[4] W[2, 4, 7, 9, 10] -
t[1] t[2] t[4] W[2, 4, 7, 9, 10] - t[2] t[4] W[2, 5, 6, 7, 10] +
t[2] t[4] W[2, 5, 7, 8, 10] - t[1] t[2] t[4] W[2, 5, 7, 8, 10] - t[2] t[4] W[2, 5, 7, 9, 10] +
t[1] t[2] t[4] W[2, 5, 7, 9, 10] + t[2] t[4] W[2, 6, 7, 9, 10] - t[2] t[4] W[2, 7, 8, 9, 10] +
t[1] t[2] t[4] W[2, 7, 8, 9, 10] + t[2] t[4] W[3, 4, 5, 6, 8] - t[2] t[4] W[3, 4, 5, 7, 8] +
t[1] t[2] t[4] W[3, 4, 5, 7, 8] + t[2] t[4] W[3, 4, 5, 8, 9] - t[1] t[2] t[4] W[3, 4, 5, 8, 9] -
t[2] t[4] W[3, 4, 5, 8, 10] + t[1] t[2] t[4] W[3, 4, 5, 8, 10] - t[2] t[4] W[3, 4, 6, 8, 9] +
t[2] t[4] W[3, 4, 7, 8, 9] - t[1] t[2] t[4] W[3, 4, 7, 8, 9] - t[2] t[4] W[3, 4, 8, 9, 10] +
t[1] t[2] t[4] W[3, 4, 8, 9, 10] + t[2] t[4] W[3, 5, 6, 8, 10] -
t[2] t[4] W[3, 5, 7, 8, 10] + t[1] t[2] t[4] W[3, 5, 7, 8, 10] +
t[2] t[4] W[3, 5, 8, 9, 10] - t[1] t[2] t[4] W[3, 5, 8, 9, 10] - t[2] t[4] W[3, 6, 8, 9, 10] +
t[2] t[4] W[3, 7, 8, 9, 10] - t[1] t[2] t[4] W[3, 7, 8, 9, 10] + t[2] t[4] W[4, 5, 6, 7, 8] -
t[2] t[4] W[4, 5, 7, 8, 9] + t[1] t[2] t[4] W[4, 5, 7, 8, 9] + t[2] t[4] W[4, 5, 7, 8, 10] -
t[1] t[2] t[4] W[4, 5, 7, 8, 10] + t[2] t[4] W[4, 6, 7, 8, 9] - t[2] t[4] W[4, 7, 8, 9, 10] +
t[1] t[2] t[4] W[4, 7, 8, 9, 10] - t[2] t[4] W[5, 6, 7, 8, 10] + t[2] t[4] W[5, 7, 8, 9, 10] -
t[1] t[2] t[4] W[5, 7, 8, 9, 10] - t[2] t[4] W[6, 7, 8, 9, 10] ] }

```

In[ ]:= {1, -1}.(Last /@ res2)

Out[ ]:= 0

Just for fun, let's try that again with all crossings flipped :

```


In[ ]:= VDDFlipped = {
  CircuitDiagram[Xp[1, 2, 11, 8], Xm[11, 3, 12, 7], Xp[12, 4, 13, 10], Xm[13, 5, 6, 9]],
  CircuitDiagram[Xp[1, 4, 11, 10], Xm[11, 5, 12, 9], Xp[12, 2, 13, 8], Xm[13, 3, 6, 7]]
} /. {Xp[i_, j_, k_, L_] => Xm[L, i, j, k], Xm[i_, j_, k_, L_] => Xp[j, k, L, i]}

Out[ ]:= {CircuitDiagram[Xm[8, 1, 2, 11], Xp[3, 12, 7, 11], Xm[10, 12, 4, 13], Xp[5, 6, 9, 13]],
  CircuitDiagram[Xm[10, 1, 4, 11], Xp[5, 12, 9, 11], Xm[8, 12, 2, 13], Xp[3, 6, 7, 13]]}

In[ ]:= res3 = (pA /@ VDDFlipped);
{1, -1}.(Last /@ res3) /. {t[5] -> t[4], t[3] -> t[2]}

Out[ ]:= 0
    
```

## Double Delta



```

In[ ]:= Timing[res4 = pA[#, (t[1] == t[6]) (t[11] == t[16]) (t[21] == t[26])] & /@ {
  CircuitDiagram[
    Xm[19, 1, 20, 2], Xp[11, 3, 12, 2], Xp[3, 30, 4, 29], Xm[4, 21, 5, 22],
    Xp[6, 23, 7, 22], Xm[7, 28, 8, 29], Xm[12, 8, 13, 9], Xp[18, 10, 19, 9],
    Xm[27, 13, 28, 14], Xp[23, 15, 24, 14], Xm[24, 16, 25, 17], Xp[26, 18, 27, 17]
  ],
  CircuitDiagram[
    Xp[1, 28, 2, 27], Xm[2, 23, 3, 24], Xm[17, 3, 18, 4], Xp[13, 5, 14, 4],
    Xm[14, 6, 15, 7], Xp[16, 8, 17, 7], Xp[8, 25, 9, 24], Xm[9, 26, 10, 27],
    Xm[29, 11, 30, 12], Xp[21, 13, 22, 12], Xm[22, 18, 23, 19], Xp[28, 20, 29, 19]
  ]
}
    
```

Out[ ]:= {3.21875, ... 1 ...}

large output    show less    show more    show all    set size limit...

```

In[ ]:= Equal @@ (Last /@ res4)
    
```

```

Out[ ]:= True
    
```

```
In[ ]:= {Date[], TimeUsed[]} // Column  
Out[ ]:= {2021, 2, 22, 8, 40, 46.0076837}  
4.969
```